Three Level Production System (smod_t3)

Outline

- Introduction
- Agents in the *smod_t3* model
- Three-level production system
- Controlling the nests
  - Sets and assignment
- Production equations
  - Three levels
- Calibration
  - Three levels
- Factor market clearing
- Implications

Agents in smod_t3

- 10 commodities
- 10 activities
- 6 factors (5 natural + 1 aggregate)
- 2 households
- Government – taxes and spends
- Savings/investment – no time dimension
- Trade
  - with a single partner, ROW
  - small country assumption
  - CET and CES for each commodity

Same structure as for *smod_t/t2*

Practical CGE, 2024 © cgemod 3

3

Three-Level Production System

Practical CGE, 2024 © cgemod 4

4

cgemod

Generic Production System: Quantities



Additions

1. $FD_{l,a}$
2. σ_{32}
3. Sets *ff* and *fag*
4. Mapping sets
   - *map_va_ff*
   - *map_fag_ff*

Practical CGE, 2024 © cgemod 5

5

cgemod

Generic Production System: Prices



Additions

1. $WFA_{l,a}$
2. σ_{32}
3. Sets *ff* and *fag*
4. Mapping sets
   - *map_va_ff*
   - *map_fag_ff*

$WFA_{f,a} = WF_f * WDIST_{f,a}$

Practical CGE, 2024 © cgemod 6

6

cgemod

cgemod

Controlling the Nests

Practical CGE, 2024 © cgemod 7

7

cgemod

Sets: Three Level Production System

```
ff(sac)          All factors natural and aggregate
f(ff)            Natural factor accounts
fag(ff)          Aggregate factors
```

f and ***fag*** are complements

```
* Maps for production nesting

 map_va_ff(ff,a)       Mapping for arguments in
                            Value added level nest

 map_fagg_ff(ff,ff,a)  Mapping for arguments in
                            nests below value added level
```

Mapping sets are assigned as to from

map_va_ff(ff,a) elements in *ff* used for (*q*)va by *a*

Practical CGE, 2024 © cgemod 8

8

cgemod

Mapping Set Assignment: Excel

map_va_ff(ff,a)					
		aagr	anres	amanu	aserv
Land and Natural Resources	flnd	1	1	1	1
Capital	fcap	1	1	1	1
Aggregate Labour	flab	1	1	1	1

Case specific 'mappings': user assigned

map_fagg_ff(ff,ff,a)					
		aagr	anres	amanu	aserv
flab	fuskl	1	1	1	1
flab	fsklb	1	1	1	1

GAMS is indifferent about 'mappings':

user must apply economic 'logic'

Garbage in, garbage out

9

Production Equations

10

cgemod

Top Level CES Equations

No changes from *smod_t2*

$$PX_a = \sum_c ioqxacqx_{a,c} * PXC_c$$

$$PX_a * (1 - TX_a) * QX_a = (PVA_a * QVA_a) + (PINT_a * QINT_a)$$

$$PINT_a = \sum_c (ioqtdqd_{c,a} * PQD)_c$$

$$ADX_a = [(adxb_a + dabadx_a) * ADXADJ] + (DADX * adx01_a)$$

$$QX_a = AD_a^x \left(\delta_a^x QVA_a^{-rhoc_a^x} + (1 - \delta_a^x) QINT_a^{-rhoc_a^x}\right)^{-\frac{1}{rhoc_a^x}} \quad \forall aqx_a$$

$$\frac{QVA_a}{QINT_a} = \left[\frac{PINT_a}{PVA_a} * \frac{\delta_a^x}{(1 - \delta_a^x)}\right]^{\frac{1}{(1+rhoc_a^x)}} \quad \forall aqx_a$$

Practical CGE, 2024 © cgemod 11

11

cgemod

Top Level CES Equations - GAMS

No changes from *smod_t2*

```
PXDEF(a)..          PX(a)  =E= SUM(c,ioqxcqx(a,c)*PXC(c)) ;

PVADEF(a)..
  PX(a)*(1-TX(a))*QX(a)  =E= PVA(a)*QVA(a))+(PINT(a)*QINT(a)) ;

PINTDEF(a)..
   PINT(a)  =E= SUM(c,ioqtdqd(c,a) * PQD(c)) ;

ADXEQ(a)..
    ADX(a)  =E=  ((adxb(a)+dabadx(a))*ADXADJ) + (DADX*adx01(a)) ;

QXPRODFN(a)$aqx(a)..
 QX(a)  =E= ADX(a)*(deltax(a)*QVA(a)**(-rhocx(a))
         + (1-deltax(a))*QINT(a)**(-rhocx(a)))**(-1/rhocx(a)) ;

QXFOC(a)$aqx(a)..
     QVA(a)  =E= QINT(a)*((PINT(a)/PVA(a))
                 *(deltax(a)/ (1-deltax(a))))**(1/(1+rhocx(a))) ;
```

Practical CGE, 2024 © cgemod 12

12

cgemod

Top Level Leontief Equations

No changes from *smod_t2*

$$QINT_a = ioqintqx_a * QX_a \qquad \forall aqx_a$$

```
QINTDEF(a)$aqxn(a)..
   QINT(a) =E= ioqintqx(a) * QX(a) ;
```

$$QVA_a = ioqvaqx_a * QX_a \qquad \forall aqxn_a$$

```
QVADEF(a)$aqxn(a)..
   QVA(a) =E= ioqvaqx(a) * QX(a) ;
```

aqxn(a) complement to aqx(a)

```
* Intermediate Input Demand
```

$$QINTD_c = \sum_a ioqtdqd_{c,a} * QINT_a$$

```
QINTDEQ(c)..
   QINTD(c) =E= SUM(a,ioqtdqd(c,a)
                 *QINT(a)) ;
```

```
* Commodity Output
```

$$QXC_c = \sum_a ioqxac_{a,c} * QX_a$$

```
COMOUT(c)..
   QXC(c) =E= SUM(a,ioqxcqx(a,c)
                 *QX(a)) ;
```

Practical CGE, 2024 © cgemod 13

13

Second-Level Equations

$$ADVA_a = \left[\left(advab_a + dabadva_a \right) * ADVAADJ \right] + \left(DADVA * adva01_a \right)$$

Indexing on *ff*

$$QVA_a = AD_a^{va} * \left[\sum_{ff\$\left[map\_va\_ff_{ff,a} \text{ and } \delta_{ff,a}^{va}\right]} \delta_{ff,a}^{va} * ADFD_{ff,a} * FD_{ff,a}^{-\rho_a^{va}} \right]^{-1/\rho_a^{va}} \qquad \forall \rho_a^{va}$$

$$WF_{ff} * WFDIST_{ff,a}$$

$$= PVA_a * AD_a^{va} * \left[\sum_{ff\$\delta_{ff,a}^{va}} \delta_{ff,a}^{va} * ADFD_{ff,a} * FD_{ff,a}^{-\rho_a^{va}} \right]^{-\left(\frac{1+\rho_a^{va}}{\rho_a^{va}}\right)} * \delta_{ff,a}^{va} * FD_{ff,a}^{\left(-\rho_a^{va}-1\right)}$$

$$= PVA_a * QVA_a * AD_a^{va} * \left[\sum_{ff\$\delta_{ff,a}^{va}} \delta_{ff,a}^{va} * ADFD_{ff,a} * FD_{ff,a}^{-\rho_a^{va}} \right]^{-1}$$

$$* \delta_{ff,a}^{va} * ADFD_{ff,a}^{-\rho_a^{va}} * \delta_{ff,a}^{va} * FD_{ff,a}^{\left(-\rho_a^{va}-1\right)}$$

$$\forall \delta_{f,a}^{va} \text{ and } map\_va\_ff_{ff,a}$$

Practical CGE, 2024 © cgemod 14

14

cgemod

Second-Level Equations - GAMS

```
ADVAEQ(a)..
 ADVA(a) =E= ((advab(a)+dabadva(a))*ADVAADJ)+(DADVA*adva01(a)) ;
```

Indexing on *ff*

```
QVAPRODFN(a)$rhocva(a)..
 QVA(a) =E= ADVA(a)
           *(SUM(ff$[map_va_ff(ff,a) AND deltava(ff,a)],
             deltava(ff,a)*[ADFD(ff,a)*FD(ff,a)]
                 **(-rhocva(a))))
                         **(-1/rhocva(a)) ;

QVAFOC(ff,a)$[map_va_ff(ff,a) AND deltava(ff,a)]..
 WF(ff) * WFDIST(ff,a)
         =E= PVA(a)*QVA(a)
              * {SUM[ffp$deltava(ffp,a),deltava(ffp,a)
              * [ADFD(ffp,a)*FD(ffp,a)]**(-rhocva(a))]}**(-1)
              * deltava(ff,a)*ADFD(ff,a)**(-rhocva(a))
                     *FD(ff,a)**(-rhocva(a)-1) ;
```

Practical CGE, 2024 © cgemod 15

15

cgemod

Third-Level Equations

$$ADFAG_{ff,a} = \left(adfagb_{ff,a} + dabfag_{ff,a}\right) + \left(ADFAGfADJ_{ff} * ADFAGaADJ_{a}\right)$$

Indexing on *ff*

map_fagg_ff(ff,ffp,a) on LHS and RHS

$$FD_{ff,a} = ADFAG_{ff,a} * \left(\sum_{ffp\$\left[\delta^{fd}_{ff,ffp,a} \text{ and } map\_fagg\_ff_{ff,ff,a}\right] f} \delta^{fd}_{ff,ffp,a} * \left(FD_{ffp,a}\right)^{\rho^{fd}_{ff,a}} \right)^{\left(-1/\rho^{fd}_{ff,a}\right)}$$

$$\forall \sum_{ffp} map\_fagg\_ff_{ff,ffp,a}$$

$$\delta^{fd}_{ff,ffp,a}$$

$$WF_{ff} * WFDIST_{ffp,a} =$$

$$WF_{ff} * WFDIST_{ff,a} * FD_{ff,a}$$

$$* \left[\sum_{ffpp\$\left[\delta^{fd}_{ff,ffpp,a} \text{ and } map\_fagg\_ff_{ff,ffpp,a}\right]} \delta^{fd}_{ff,ffpp,a} * FD_{ffpp,a}^{-\rho^{fd}_{ff,a}} \right]^{(-1)} * \delta^{fd}_{ff,ffp,a} * FD_{ffp,a}^{\left(-\rho^{fd}_{ff,a}-1\right)}$$

$$\forall \delta^{fd}_{ff,ffpp,a}, map\_fagg\_ff_{ff,ffpp,a}$$

Practical CGE, 2024 © cgemod 16

16

cgemod

Third-Level Equations - GAMS

```
FDPRODFN(ff,a)$SUM[ffp,map_fagg_ff(ff,ffp,a)]..
 FD(ff,a) =E= ADFAG(ff,a)
             *{SUM[ffp${map_fagg_ff(ff,ffp,a)
                        AND deltafd(ff,ffp,a)},
               deltafd(ff,ffp,a) *FD(ffp,a)**(-rhofd(ff,a))]}
                       **[-1/rhofd(ff,a)] ;

* FOC for ffpp used to produce aggregate ffp by activity a

FDFOC(ff,ffp,a)$[map_fagg_ff(ff,ffp,a) AND deltafd(ff,ffp,a)]..
 WF(ffp)*WFDIST(ffp,a)
   =E= {WF(ff) * WFDIST(ff,a)*FD(ff,a)}
        *{SUM[ffpp${map_fagg_ff(ff,ffpp,a)
                   AND deltafd(ff,ffpp,a)},
          deltafd(ff,ffpp,a)*FD(ffpp,a)**(-rhofd(ff,a))]}**(-1)
              *deltafd(ff,ffp,a)*FD(ffp,a)**[-rhofd(ff,a)-1] ;

WF.FX(ff)$fag(ff)       = WF0(ff) * CPI.L ;
```

Aggregate price adjustment through $WFDIST_{ff,a}$

17

Calibration

18

Calibration: Value of Factors

```
* Value of natural factors

 V_FD0(f,a)    = SAM(f,a) ;

* Value of aggregate factors

 V_FD0(fag,a)  = SUM[ff$map_fagg_ff(fag,ff,a),SAM(ff,a)] ;
```

Why use initial 'value' parameters?

19

Calibration: Initial values for variables

No changes from *smod_t2*

```
QX0(a)                      = SAM("total",a)/PX0(a) ;
QXC0(c)                     = SUM(a,SAM(a,c))/PXC0(c) ;

QINT0(a)$SUM(c,SAM(c,a))= SUM(c,SAM(c,a)/PQD0(c)) ;

QVA0(a)                     = SUM(f,SAM(f,a)) ;

PINT0(a)$(QINT0(a))  = SUM(c,(SAM(c,a)/PQD0(c)/QINT0(a))*PQD0(c));

PVA0(a)$QVA0(a)             = SUM(f,SAM(f,a))/QVA0(a) ;

QINTD0(c)$SUM(ap,SAM(c,ap))  = SUM(a,SAM(c,a)/PQD0(c)) ;

FD0(f,a)                    = FACTUSE(f,a) ;
FS0(f)                      = SUM(a,FACTUSE(f,a)) ;

WF0(f)$(FS0(f))             = SUM(a,SAM(f,a))/FS0(f) ;

WFDIST0(f,a)$FD0(f,a)       = (SAM(f,a)/FD0(f,a))/WF0(f) ;
WFDIST0(f,a)$(FD0(f,a) EQ 0)          = 0.0 ;
```


20

Calibration: Initial values aggregate factors

***f* and *fag* are complements**

```
FD0(fag,a)  = V_FD0(fag,a) ;

Parameter
 neg_FD0(ff,a)    Identify neg_FD0
 CHK_neg_FD0      CHECK on neg_FD0  ;

neg_FD0(ff,a)$(FD0(ff,a) LT 0.0) = FD0(ff,a) ;
CHK_neg_FD0 = SUM[(ff,a)$neg_FD0(ff,a), 1 ] ;
ABORT $(CHK_neg_FD0 GT 0.0)
  "At least ONE factor demand (FD0) is negative" ;

FS0(fag)     = SUM[a,FD0(fag,a)] ;

WF0(fag)$SUM[a, FD0(fag,a)] = SUM[a,V_FD0(fag,a)]
                                /SUM[a, FD0(fag,a)] ;
* WF0(fag)$[FS0(fag)] = 1.0 ;

WFDIST0(fag,a)$FD0(fag,a) = {V_FD0(fag,a)/FD0(fag,a)}/WF0(fag) ;
WFDIST0(fag,a)$[FD0(fag,a) EQ 0] = 0.0 ;
```

Practical CGE, 2024 © cgemod 21

21

Calibration: Level 1 CES

No changes from *smod_t2*

```
rhocx(a)     = (1/ELASTX(a,"sigmax")) - 1 ;
```

ELASTX(a,"sigmax") - **exogenous**

```
predeltax(a)$(QINT0(a))
                     = (PVA0(a)/PINT0(a))*(QVA0(a)/QINT0(a))
                       **(1+rhocx(a)) ;

deltax(a)     = predeltax(a)/(1.0+predeltax(a)) ;

ADX0(a)$deltax(a)
    = QX0(a)/(deltax(a)*QVA0(a)**(-rhocx(a))
      +(1-deltax(a))*QINT0(a)**(-rhocx(a)))**(-1/rhocx(a)) ;
```

Other values only use TVs

```
ADXADJ0       = 1 ;
DADX0         = 0 ;
adxb(a)       = ADX0(a) ;
dabadx(a)     = 0.0 ;
adx01(a)      = 0.0 ;
```

Practical CGE, 2024 © cgemod 22

22

Calibration: Level 1 Leontief

No changes from *smod_t2*

```
use(c,a)$SAM("TOTAL",a)          = SAM(c,a)/SAM("TOTAL",a) ;
comactco(c,a)$(QX0(a)$PQD0(c)) = (SAM(c,a)/PQD0(c))/QX0(a) ;

ioqintqx(a)$QX0(a)    = QINT0(a)/QX0(a) ;

ioqvaqx(a)$QX0(a)     = QVA0(a)/QX0(a) ;

* Intermediate Input Demand

ioqtdqd(c,a)$(QINT0(a)$PQD0(c))
                      = SAM(c,a)/PQD0(c)/QINT0(a) ;

* Activity Output

ioqxcqx(a,c)$acx(a)   = SAM(a,c)/SUM(cp,SAM(a,cp)) ;
```


23

Calibration: Level 2 CES

```
rhocva(a)    = (1/ELASTX(a,"sigmava")) - 1 ;
```

ELASTX(a,"sigmava") - **exogenous**

```
deltava(ff,a)$[map_va_ff(ff,a) AND FD0(ff,a)]
   = (WFDIST0(ff,a)*WF0(ff)*(FD0(ff,a))**(1+rhocva(a)))
        /SUM(ffp$map_va_ff(ffp,a),
             WFDIST0(ffp,a)*WF0(ffp)*(FD0(ffp,a))
             **(1+rhocva(a))) ;

ADVA0(a)$SUM(ff$map_va_ff(ff,a),deltava(ff,a)*FD0(ff,a))
   = QVA0(a)/(SUM(ff$map_va_ff(ff,a),deltava(ff,a)
             *FD0(ff,a)**(-rhocva(a))))**(-1/rhocva(a));

ADVAADJ0       = 1 ;
DADVA0         = 0 ;
advab(a)       = ADVA0(a) ;
dabadva(a)     = 0.0 ;
adva01(a)      = 0.0 ;
ADFD0(ff,a)    = 1 ;
```


24

Calibration: Level 2 CES CHECKS

Why such detailed checks?

```
deltavaCHK(a)  = [SUM(ff,deltava(ff,a))] - 1.0 ;

deltavaCHK(a)$[ABS(deltavaCHK(a)) GT 0.0000000000000] = 0.00 ;

deltava_neg(ff,a)$[deltava(ff,a) LT 0.000]  = 1.00 ;
```

Use of ABORT with "message" (works on scalars)

```
ABORT $(SUM[a,deltavaCHK(a)] NE 0.0)
 "SUM of ONE OR MORE deltava is not ONE
        - view deltava and deltavaCHK" ;

ABORT $(SUM[(ff,a),deltava_neg(ff,a)] NE 0.0)
 "ONE OR MORE deltava is negative
         - view deltava and deltava_neg" ;
```

Practical CGE, 2024 © cgemod 25

25

Calibration: Level 3 CES

```
rhofd(fag,a)   = (1/ELASTF(fag,a)) - 1 ;
```

ELASTF(ff,a) - **exogenous**

Shares of ffp to produce aggregate ff by a

```
deltafd(ff,ffp,a)${map_fagg_ff(ff,ffp,a) AND
               SUM[ffpp$map_fagg_ff(ff,ffpp,a), FD0(ffpp,a)] }
      = {[WF0(ffp)*WFDIST0(ffp,a)]*(FD0(ffp,a))
                     **[1+rhofd(ff,a)]}
         /{SUM(ffpp$map_fagg_ff(ff,ffpp,a),
               [WF0(ffpp)*WFDIST0(ffpp,a)]*(FD0(ffpp,a))
                      **[1+rhofd(ff,a)])} ;

 adfag(ff,a)${SUM[ffp,map_fagg_ff(ff,ffp,a)] AND
              SUM[ffp$map_fagg_ff(ff,ffp,a),FD0(ffp,a)]  }
      = FD0(ff,a)/{ SUM[ffp$map_fagg_ff(ff,ffp,a),
           deltafd(ff,ffp,a)*FD0(ffp,a)**(-rhofd(ff,a))] }
                  **[-1/rhofd(ff,a)] ;

 adfag0(ff,a) = adfag(ff,a) ;
```

Practical CGE, 2024 © cgemod 26

26

Calibration: Level 3 CES CHECKS

```
Parameter
test_deltafd(ff,a)  Denominator for deltafd calculation ;
test_deltafd(ff,a)  ={SUM(ffpp$map_fagg_ff(ff,ffpp,a),
                        [WF0(ffpp)*WFDIST0(ffpp,a)]*(FD0(ffpp,a))
                          **[1+rhofd(ff,a)])} ;

deltafdCHK(ff,a)$fag(ff)
  = SUM(ffp$map_fagg_ff(ff,ffp,a),deltafd(ff,ffp,a)) - 1.0 ;

deltafdCHK(ff,a)$[ABS(deltafdCHK(ff,a)) GT 0.00000000] = 0.00 ;

deltafd_neg(ff,ff,a)$[deltafd(ff,ff,a) LT 0.000]  = 1.00 ;

ABORT $(SUM[(ff,a),deltafdCHK(ff,a)] NE 0.0)
 "SUM of ONE OR MORE deltafd is not ONE
              - view deltafd and deltafdCHK" ;

ABORT $(SUM[(ff,ffp,a),deltafd_neg(ff,ffp,a)] NE 0.0)
 "ONE OR MORE deltafd is negative
               - view deltafd and deltafd_neg" ;
```

Practical CGE, 2024 © cgemod 27

27

Calibration: Reporting Nests

```
* Reporting the nesting structure

nest_va(ff,a)$[deltava(ff,a) GT 0.0]       = 1.0 ;
nest_fd(ff,ffp,a)$[deltafd(ff,ffp,a) GT 0]= 1.0 ;
```

Practical CGE, 2024 © cgemod 28

28

cgemod

Factor Market Clearing

cgemod

Factor Market Clearing

Relative to *smod_t2*

```
* Technology for factor & activity specific factor efficiency

ADFD.FX(f,a)=ADFD0(f,a);  →  ADFD.FX(ff,a)=ADFD0(ff,a);

  * NOT a closure since always fixed (but)

  WF.FX(ff)$fag(ff) = WF0(ff) * CPI.L ;
```

Aggregate price adjustment is through $WFDIST_{ff,a}$

cgemod

cgemod

Implications

31

cgemod

Implications 1

- PVA_a changes relative to $PINT_a$
  - Induces changes in QVA_a and $QINT_a$
- PQD_c changes for one or more intermediate
  - Induces changes in $PINT_a$
- $ADVA_a$ changes
  - Induces changes in PVA_a
- $ADFD_{ff,a}$ changes for one or more (*ff*,*a*)
  - Induces changes in
    - $WF_{ff}*WFDIST_{ff,a}$
    - PVA_a

ELASTX(a,“sigmax”/”sigmava”) and deltax/deltava determine responsiveness

Leontief implies sigma = 0.0

32

cgemod

Implications 2

- WFA_{fag} changes relative to WFA_{ff}
  - Induces changes in input mix to QVA_{a}
- $FDPRODN_{'fag',a}$ allows changes input mix for *fag* 'independent' of other 'natural' factors
  - Responsiveness determined by $ELASTFD_{fag,a}$
- $ADFAG_{fag,a}$ changes
  - Increases flow of services from $FD_{fag,a}$ (Hicks neutral)
- $ADFD_{f,a}$ changes for one or more (f,a)
  - Induces changes in $WFDIST_{fag,a}$

Why might we assume that skilled and unskilled are as substitutable for each other as they are for capital and/or land?

Practical CGE, 2024 © cgemod 33

33

The End

Three Level Production System (*smod_t3*)

Practical CGE, 2024 © cgemod 34

34